
Abstract 
This paper will show that a person can be new to SAS and create beneficial reports. Armed with 
the SAS Basic Programming Essentials Training Manual and using the ODS and ExcelXP Tagset, 
data results will be moved from SUDAAN to SAS and finally into EXCEL without touching the data, 
using the names of the risk factor variables, or how many values those variables had. The jumbled 
SUDAAN output will be transformed into something that is informative and easy to read. And when 
my supervisor wanted charts with confidence intervals, the simple solution was incorporating a 
chart template into the syntax! 

Introduction 
“The Texas Behavioral Risk Factor Surveillance System (BRFSS), initiated in 1987, 
is a federally funded (CDC) telephone survey conducted on a monthly basis of 
randomly selected adult Texans to collect data on lifestyle risk factors contributing to 
the leading causes of death and chronic diseases. As a primary source for 
comprehensive statewide data on preventive health practices and health risk 
behaviors, BRFSS is an important tool for decision-making throughout DSHS and the 
public health community. Public and private health authorities at the federal, state, 
and local levels rely on BRFSS to identify public health problems, design policies and 
interventions, set goals, and measure progress toward those goals.” 

  BRFSS webpage: http://www.dshs.state.tx.us/chs/brfss/default.shtm  
 
Since the BRFSS data touches almost every program in the state health department, data needs 
will always be different based on the program requesting it. Programs may want one or all of the 
standard risk factors and health indicators for one or more counties, or they might want different 
aggregations of age groups, education groups, or income groups.  In addition, sometimes the way 
a question is asked will change from time to time, or the answers to those questions will change. 
For example, the question on diabetes changed from two answers to four answers. The BRFSS is 
very fluid and dynamic, and the results produced from it need to be as well. 
 
Typically, data results are obtained in either SUDAAN or SPSS and then hand typed or cut and 
pasted into pre-designed tables. This is time consuming and prone to errors. A new solution was 
needed that was fast and readable for clients, to make reports that had titles, footnotes, 
confidence intervals, demographics and current dates for sometimes up to 130 tables! 

Getting Started 

We start in SUDAAN as they have the “gold standard” for confidence intervals (CI) that the 
Centers for Disease Control (CDC) desire. SAS 9.2 currently gives only symmetrical CI so we 

http://www.dshs.state.tx.us/chs/brfss/default.shtm


SCSUG Educational Forum, 2011 

 2 

must start in SUDAAN which produces asymmetrical CIs and is easier to export into SAS than 
SPSS.  
 
This is the code that we use: 
 

proc crosstab data = "H:\brfss_data\brfss09\state\state_09_Working_av.sav" design = wr filetype= spss  

 nowgt nocol; 

nest STSTR seqno/ missunit; 

weight cdcfinwt; 

setenv linesize = 106 pagesize = 45 colwidth = 11 topmgn = 1; 

subgroup agegr4 raceeth sex educat4 inccat3b admin2 fairpoor keptusal c06q01 diabetes rfsmok  

 c03q04 chckup1yr ; 

levels    4 4  2  4 3  8    2  2  4  2  2 8

 2  ;  

tables  admin2*(fairpoor keptusal c06q01 diabetes rfsmok c03q04 chckup1yr)*(sex raceeth agegr4  

 educat4  inccat3b); 

print nsum rowper lowrow uprow; 

output nsum rowper lowrow uprow /  

 filename="H:\Brfss\avincent\Mypresentations\master"  

 filetype=SASXPORT replace 

 Levfile="H:\Brfss\avincent\Mypresentations\mastrlev"; 

 

Texas BRFSS data files are currently managed in SPSS. With the subgroup it is important that 
known demographics need to come first, the geographic area second (if necessary) and then all 
risk factor variables to follow. Near the bottom are the few lines of code that will export the tables 
into SAS. The Levfile is what your format file is named.  
 
From SUDAAN: “LEVFILE=filename: Optional parameter, standalone SUDAAN versions only.  
Use LEVFILE=filename with SASXPORT input files to specify the name of an optional file 
containing records with SAS-style format information.  If this parameter is not supplied, SUDAAN 
will look for a file named LEVFILE.STX.  If SUDAAN cannot find a level file, whether the name is 
supplied explicitly or by default, then SUDAAN will not be able to use formats named on FORMAT 
statements within your SUDAAN program. This option is available in all SUDAAN Procedures.“ 
 
After the SUDAAN program has been run, the data file in SAS looks like this (Picture 1): 
 
Picture 1: 

The columns for the demographics are in the same order as they are in the subgroup for 
SUDAAN. Then SUDAAN runs the data as you have your tables set up in SUDAAN. So the 
values that are created for each column are created in the order that the variables are in the 
tables section. In this example the values that are created first are admin2 by sex by fairpoor. 
 
Also notice the _C columns; these will be used for arrays. I do not know if SUDAAN or SAS 
creates these, but they can be very useful. 
 
To help decipher the output here is an example of what is wanted: 
 
Picture 2: 

 
These are the variables that are needed to get the grand totals for the “male” demographic of the 
health risk variables. In this case FAIRPOOR was created first because it was first in the table 
listing on SUDAAN. It is also the first question that is asked on the survey. “Would you say that in 
general your health is…Excellent, Very good, Good, Fair, or Poor?”. 
 
The SEX column will also be used for our grand totals or sample size, NSUM, for the area variable 
(ADMIN2), in this case Public Health Region 2/3. The reason why the SEX column is used for 
totals is that gender is logged by the interviewer and there are no missing or refused answers. 
 
Lines 22 and 23 in Picture 2 must be on the same line on the final report. It must have titles and 
footnotes and reformatted to look like this (Picture 3): 

Picture 3: 

 

So how do you get here? 
 
Well first, one must have the ExcelXP Tagset which can be downloaded from the Support-SAS 
website: 
 

o http://support.sas.com/rnd/base/topics/odsmarkup/ 
o Scroll for the ExcelXP downloads. 
o There are some very good papers that detail how to create Excel files so be sure to have 

a look at them! 
o Once downloaded, open and copy/paste into SAS editor and run.  
o One must also have the SAS/ACCESS module/add-on. 

 

Getting into SAS 

The opening statements are how to get SUDAAN output into SAS: 
 
libname present 'H:\Brfss\avincent\MyPresentations'; 

 

libname show xport "H:\Brfss\avincent\MyPresentations\mastrlev.stx" ; 

options nofmterr; 

 

data masterF; 

set show.mastrlev; 

run; 

 

proc format cntlin=show.mastrlev; 

run; 

 

libname showmf xport "H:\Brfss\avincent\MyPresentations\master.stx";

 options nofmterr; 

 

data present.mainfile; 

http://support.sas.com/rnd/base/topics/odsmarkup/

set showmf.master; 

run; 

 

The top library opens a pathway where all the data sets and files will go to. The other two are 
export libraries that will be around long enough to give the data table. Then we can either clear the 
libraries to shut them down or just leave them be. 
 
We then run some proc statements to see the contents of the data file. 
 
proc print data=present.mainfile (obs=50) ; 

run; 

 

proc contents data=present.mainfile position; 

run; 

 

The proc print you have seen in the pictures above. This is the important part of the proc contents: 
 
Picture 4: 

 
 

This shows you if either SUDAAN or SAS truncated your variable names and how many _C 
columns you have. 

Next, we set up our Macros! 
 
%macro fdate(fmt); 

   %global fdate; 

   data _null_; 

      call symput("fdate",left(put("&sysdate"d,&fmt))); 

   run; 

%mend fdate; 

%fdate(worddate.) 

 

%let title1 = 'Prevalence of #byval(riskv)'; 

%let title2 = '#byval(area)'; 

%let title3 = 'Adults Ages 18 Years and Over'; 

%let title4 = 'Texas BRFSS, 2010'; 

%let footnote1 = 'Survey Question: '; 

%let footnote3 = 'Note: All reported rates are weighted for Texas 

 Demographics and the probability of selection.'; 

%let footnote4 = 'Note: Prepared using complex sample design from 

 SUDAAN'; 

%let footnote5 = 'Prepared by:'; 

%let footnote6 = 'Texas Behavioral Surveillance System, Center for 

 Health Statistics'; 

%let footnote7 = 'Texas Department of State Health Services'; 

 

The date macro will date the tables when they are run and put the date into a word format like 
“January 19, 2011”. The other macros are for the titles and footnotes. These are easier to have at 
the beginning of the syntax so that they can be modified for each new run. 
 
The “#byval()” macro (which can be found on title1 and title2 lines) is utilized in the proc report 
option “by”. This allows for variable data to be added into the report titles or footnotes. This is 
especially nice when 130 tables need to be run! 
 
The next part of the syntax is formatting the variables. Again, this is added near the beginning of 
the program where it can be easily changed based on the risk factors or area of interest. It is also 
a good idea to keep a list from SUDAAN of what risk factors were used and in what order they 
were in the subgroup heading.  
 
proc format; 

 value riskf  1 = 'General Health Fair to Poor' 

2 = 'Kept from Usual Activities due to Poor 

Mental or Physical Health' 

    3 = 'Diabetes - All Answers' 

    4 = 'Diabetes - Yes/No' 

    5 = 'Current Smoker' 

    6 = 'Routine Checkup - All Answers' 

    7 = 'Routine Checkup Within the Last Year' 

    ; 

 

 value areaf    1 = 'PHR 1' 

     2 = 'PHR 2/3' 

     3 = 'PHR 4/5N' 

     4 = 'PHR 6/5S' 

     5 = 'PHR 7' 

     6 = 'PHR 8' 

     7 = 'PHR 9/10' 

     8 = 'PHR 11' 

     ; 

run; 

The Real Work 

The program that was created is broken down by each section with an explanation of why it was 
done. 
 
Creating Groups and Formatting 
 

Section 1: 
 
Data present.data1; 

 set present.mainfile; 

 

 rename admin2=area;  

 if (agegr4=0 or raceeth=0 or educat4=0 or inccat3b=0 or admin2=0) 

  then delete; 

 length grp 3 grp2 $3; 

 if sex=0 then grp=1; 

 else if sex > 0 then grp=2; 

 else if raceeth > 0 then grp=3; 

 else if agegr4 > 0 then grp=4; 

 else if educat4 > 0 then grp=5; 

 else if inccat3b > 0 then grp=6; 

 

In the above, the variable admin2 is renamed into something that will be used for the rest of the 
syntax. That way only one area variable needs to be modified when a new data request comes in 
(e.g., admin2 would be replaced with region). All demographic totals are removed except for “sex” 
because gender has no missing values. A new variable called “grp” was created so that the 
groups can be resorted in the order desired after the sorting and transposing of data results. 
 
Section 2: 
 
 Length alltitle $15 subtitle $16 ; 

 if grp=1 then alltitle="Totals"; 

 else if grp=2 then alltitle="Gender"; 

 else if grp=4 then alltitle="Age Group"; 

 else if grp=3 then alltitle="Race/Ethnicity"; 

 else if grp=5 then alltitle="Education"; 

 else if grp=6 then alltitle="Income"; 

  

 if agegr4=1 then subtitle='18-29 Years'; 

 else if agegr4=2 then subtitle='30-44 Years'; 

 else if agegr4=3 then subtitle='45-64 Years'; 

 else if agegr4=4 then subtitle='65+ Years'; 

 else if sex=1 then subtitle='Male'; 

 else if sex=2 then subtitle='Female'; 

 else if raceeth=1 then subtitle='White'; 

 else if raceeth=2 then subtitle='Black'; 

 else if raceeth=3 then subtitle='Hispanic'; 

 else if raceeth=4 then subtitle='Other'; 

 else if educat4=1 then subtitle='< High School'; 

 else if educat4=2 then subtitle='High School Grad'; 

 else if educat4=3 then subtitle='Some College'; 

 else if educat4=4 then subtitle='College Grad'; 

 else if inccat3b=1 then subtitle='< $25,000'; 

 else if inccat3b=2 then subtitle='$25,000-$49,999'; 

 else if inccat3b=3 then subtitle='$50,000 +'; 

 else if alltitle='Totals' then subtitle='Totals'; 

 

In section 2, the variables and demographic groups are reformatted the way they should look like 
in the final report. 
 
Section 3:  
 
 if subtitle = 'Totals' then add1='a'; 

 if subtitle = 'Male' then add1='b'; 

 if subtitle = 'Female' then add1='c'; 

 if subtitle = 'White' then add1='d'; 

 if subtitle = 'Black' then add1='e'; 

 if subtitle = 'Hispanic' then add1='f'; 

 if subtitle = 'Other' then add1='g'; 

 if subtitle = '18-29 Years' then add1='h'; 

 if subtitle = '30-44 Years' then add1='i'; 

 if subtitle = '45-64 Years' then add1='j'; 

 if subtitle = '65+ Years' then add1='k'; 

 if subtitle = '< High School' then add1='l'; 

 if subtitle = 'High School Grad' then add1='m'; 

 if subtitle = 'Some College' then add1='n'; 

 if subtitle = 'College Grad' then add1='o'; 

 if subtitle = '< $25,000' then add1='p'; 

 if subtitle = '$25,000-$49,999' then add1='q'; 

 if subtitle = '$50,000 +' then add1='r'; 

 grp2=trim(add1); 

 

Section 3 shows a way of keeping the demographics in the order desired. These are alphabetic 
because the first variable “grp” is numeric and will make it easier to tell them apart. 
 
Section 4: 
 
LCL = Round(lowrow, .1); 

if LCL < 0 then LCL = 0; 

UCL = Round(uprow, .1); 

if UCL < 0 then UCL = 0; 

 

Length CI $15 Value $20; 

CI ="(" || Compress(Put(LCL, 5.1))||"-"||Compress(Put(UCL, 5.1))|| ")"; 

Value = Put(Round(rowper, .1), 5.1) || " " || CI; 

Run; 

 

In Section 4 of the program, new names are given to the upper and lower CIs and are rounded to 
one decimal place. In the second part, parentheses are added to the “ci” variable as well as a 
dash between them, and then the “ci” is placed behind the estimated percentage which is also 
rounded to one decimal place. In the end it will look like this: 12.1 (9.5-15.3). 
 
Section 5: 
 

data present.arrayrf ; 

 set present.data1; 

 array rfactor(*) _c7 -_c50 ; 

 do i=1 to dim(rfactor); 

  if rfactor(i)=> 0 then rfactor(i)=(i); 

   else rfactor(i)=.; 

 rf = SUM(of _c7 - _c50) ; 

 end; 

   

run; 

 

An array is setup for section 5 of the programming. This is where the _C columns are utilized. The 
risk factor data results start in column _c7; this is why all of the demographics (in this case five) 
must be first in the subgroup section of SUDAAN and the area second (making six which is why 
we start in _c7).  
 
This array is numbering the risk factors and putting them into one column. So, in the variable ‘rf’, 
_c7 will be 1, _c8 is 2, etc. down to however many risk factors there are. If the _c column is empty 
then the counting will stop. 
 
This is what the data set looks like: 
 
Picture 5: 

 
Section 6: 
 
data present.data2; 

 set present.arrayrf; 

 if rowper = 100 then value = .; 

 if rowper = 0 then nsum = .; 

 if lowrow > 0 then nsum = .; 

 if uprow > 0 then nsum = .; 

  

drop _c1-_c50 procnum lcl ucl ci i; 

run; 

 

The grand total sample size of the demographics is needed. Section 6 of the program sets the 
“value” variable as missing if it is the total sample size. Only the NSUM value associated with the 
ROWPER = 100 is needed and the other NSUM values should be deleted. 
 
Variables that are no longer needed are dropped as well. 
 
 
Here is that data set cleaned up. 

Picture 6: 

 
Transposing 
 
Two data sets will need to be created, one for total sample size and one for the estimates or 
values. 
 
Section 7: 
 
proc sort data=present.data2; 

 by area rf grp grp2 alltitle subtitle; 

run; 

 

proc transpose data=present.data2 

     out=present.datatran let; 

 var  nsum value ; 

 by area rf grp grp2 alltitle subtitle ; 

run; 

 

Data must be sorted by the variables that will be used in the proc transpose step. In this case six 
variables need to be used.  
 
This is what this data set looks like: 

Picture 7: 

 
All the values are on the same row. From lines 74 on, the risk factor variables with more than two 
answers are all on the same row and the order that the values are in, is the order they are 
supposed to be in. 
 
The file will now be split into two separate files. 
 
Sample Size Data File 
 
The next data step has to do with the Texas BRFSS only reporting items that have a sample size 
of 50 or more. This is due to statistical reliability and confidentiality purposes when the respondent 
numbers are low. 
 
data present.subdata1(rename=(col1=ssizec)); 

   length ssize 8  ; 

   set present.datatran; 

   ssize=put(col1,20.); 

 if (ssize=. or ssize=0) then delete; 

 if (ssize < 50) then col1='< 50'; 

   keep area ssize col1 grp2 rf alltitle subtitle ; 

run; 

 

A trick about renaming variables. RENAME applies to all output data sets only. Use the old 
variable name in the programming statements for the current data steps. In other words, the 
rename only applies to the final data set, not the input data set that the variables come from. So 
don’t apply the new name in the data step. 
 
The ‘put’ statement transforms a character variable, (col1), to a numeric variable, (ssize), which 
will be needed for statistical computations. 
 
The ‘value’ columns are removed and a data set with only the total sample size is established. 
 
Here’s that data set. 

Picture 8: 

 
 

We see the variable renamed from “col1” to “ssizec” and now has ‘< 50’ value from “ssize” being 
under 50 count.  
 
Values (Estimates) Data File 
 
Section 9: 
 
The second data file will have the sample sizes removed and only the values left. 
 
data present.subdata2 ; 

   set present.datatran; 

 if (_NAME_='NSUM') then delete; 

keep area grp2 rf alltitle subtitle col2-col50; 

run; 

 
Picture 9: 

 
 

Section 10: 
 
The two data files now have to be merged back together again so the data are all on one line. 
 
proc sort data=present.subdata1; 

 by area rf grp2; 

run; 

 

proc sort data=present.subdata2; 

 by area rf grp2; 

run; 

 

data present.data3; 

 merge present.subdata1 present.subdata2 ; 

 by area rf grp2 ; 

run; 

 

proc sort data=present.data3; 

 by area rf grp2; 

run; 

 

First the subdata sets need to be sorted by the variables that they will be matched on; in this case 
those variables are “area,” the risk factor variable “rf,” and the demographic variable “grp2.” The 
sample size data set needs to come first followed by the values data set because we want the 
sample size on the left hand side of the final table. The data are sorted one last time, but this step 
could porbably be skipped. 
 
Picture 10: 

 
 

Every item is on the same row, and it all lines up. Since the data results are not reported for those 
areas with < 50 respondents, these “values” need to be suppressed and reformatted to ‘(-)’. 
 

Section 11: 
 
data present.final; 

 set present.data3; 

 length riskv 8; 

  riskv=rf; 

 format riskv riskf. area areaf.; 

 array minusvar(*) col2-col9; 

 do i=1 to dim(minusvar); 

 if ssize < 50 then minusvar(i)='(-)'; 

 end; 

run; 

 

Another array is used to incorporate the data that are to be suppressed. For this data step the 
names of all the “_c” columns need to be known and written in. A look at the actual data table will 
be in order. 
 

Section 12: 
 
option NOBYLINE; 

 

Proc report data = present.final split = "\" nowd spacing=3 

 style(column) = [font_size = 8pt just = c] 

 headline headskip ; 

 by area riskv; 

 

Columns area riskv alltitle subtitle grp2 ssizec col2-col9 ; 

 

define area / order order =data 'Area' Noprint; 

break after area /  ol skip ; 

define riskv / order order =data 'RF'; 

break after riskv /  ol skip; 

define alltitle / order order=data 'Groups'; 

break after alltitle/ skip; 

define subtitle / 'Demographics'; 

Define grp2 / order order = data Noprint; 

break after grp2 / skip ; 

define ssizec /  'Sample' 'Size' ; 

define col2-col9 / right 'Percent' '(95% CI)';  

title1 &Title1 j=c; 

title2 &title2 j=c; 

footnote1 &footnote5 j=l;; 

run; 

 

The ‘option nobyline’ is a global option to state that the ‘by’ group should not be repeated in the 
first line after the titles. For us, this is repetitive and annoying, but it might be useful to you.  
 
The proc report is pretty straight forward. We have skips in our headlines. We have a line break 
after the area, the risk factor, and the demographic groups. This report is made just to make sure 
that all the data works and is in the correct place. 
 
This is the output: 
 
Picture 11: 

 
Notice that the titles have their risk factor and area listed correctly. This is just a quick report; the 
formatting comes later when the data results and reports are sent to Excel. 
 
Section 13: 
 
ods listing close; 

ods tagsets.excelxp file =  

           "H:\Brfss\avincent\MyPresentations\showoff.xml" 

style=statistical 

 options(sheet_label = 'Risk Factor ' 

   embedded_titles = 'yes'  

   embedded_footnotes='yes' 

   sheet_interval='table'       

   autofit_height='yes' 

   contents='yes'); 

 

ODS is versatile and easy to use. The tagset excelxp needs to be access to give the output a 
place to go. There are many styles to choose from, I use ‘statistical’. The options for ODS are 

many as well. For example, if the titles or footnotes are not embedded, they only appear when 
something is printed out. Clients typically need to have them so that they know what they are 
looking at. The ‘sheet_interval’ option tells SAS when to split the output on the EXCEL 
spreadsheet. Having the ‘contents’ option will make a table of contents spreadsheet with a link to 
every table, which is very handy when there are 130 tables! 
 
A small warning concerning options, when the “option nobyline” was used on the global statement, 
it interfered with the “sheet_interval” option. When the “option byline” was used instead for this 
report the “sheet_interval=bygroup” option worked just fine. So that is something to watch out for. 
 

Proc report data = present.final split = "\" nowd spacing=3 

 style(column) = [font_size = 10pt] 

 headline headskip ; 

 format riskv riskf. ; 

 by area riskv notsorted; 

 

In the above, the proc report is much the same as the previous one. The risk factor variable is 
formatted and the ‘by’ option has two variables. Note that the procedure has the further option of 
‘notsorted’, which means that the data set does not need to be sorted by those variables and it 
tells SAS to read the data as is.  
 

Columns area riskv alltitle subtitle grp2 ssizec col2-col9 ; 

 

define area / order order =data noprint; 

break after area /  ol skip; 

define riskv / order order =data noprint ; 

break after riskv /  ol skip; 

define alltitle / order order=data 'Groups' left width=11; 

break after alltitle/ skip; 

define subtitle / 'Demographics' left width=11; 

Define grp2 / order order = data Noprint; 

break after grp2 / skip ; 

define ssizec / style = {tagattr='format:##,###'} 'Sample\Size' right; 

define col2-col9 / center 'Percent\(95% CI)' width=11;  

 

title1 &Title1 j=c; 

title2 &Title2 j=c; 

title3 &Title3 j=c; 

title4 &Title4 j=c; 

 

footnote1 underline=1 &footnote1 j=l; 

footnote3 &footnote3 j=l; 

footnote4 &footnote4 j=l; 

footnote5 &footnote5 j=l; 

footnote6 &footnote6 j=l; 

footnote7 &footnote7 j=l; 

footnote8 "&fdate"; 

run; 

 

ods tagsets.excelxp close; 

ods listing; 

 

Most of the rest of this is pretty straight forward. The columns are listed. The area and risk factor 
variables are ‘noprint’ as they will be listed in the titles. The ssizec has a style attribute that will get 
EXCEL to change its appearance by putting in the comma in the usual numeric places.  

With the footnotes footnote2 is skipped so that there is a line space. The end of the program calls 
up the date macro that was created at the beginning of the file. 
 
The near final product in EXCEL is below: 
 
Picture 12: 

 
 

The table is not perfect, but the cleanup is easy. In addition, all we have to add is the ‘Percent’ 
answers (“Fair or Poor” or “Excellent, Very Good, Good” in this example) and the ‘Survey 
Question:’ in the footnotes.  
 
Here’s what it looks like with the less than 50 values: 
 
Picture 13: 

 
 

And here is an example of having some long answers: 

Picture 14: 

 
 

Here is something that SUDAAN does that is not the fault of SAS. Notice the three columns of all 
zeroes (middle right in Picture 14). This question asked respondents when their last routine 
medical checkup was. There are only five answers that they could give, “Within the past year”, 
“within the past two years”, “within the past 5 years”,”5 or more years ago”, or “never”. You might 
be incline to panic and say “oh no! What went wrong!?!” Well actually nothing went wrong. The 
answers to the question are numbered, 1-4 for the years, 8 = never. The other answers for this 
question, 7=”Don’t Know/Not Sure” and 9=”Refused”, are set to missing in SPSS and are not used 
for calculation. 
 
SUDAAN must have continuous numbering when it runs. So the gaps between answers 4 and 8 
must be filled by something. So when the data results are cleaned, these columns of all zeroes 
need to be deleted. Sometimes a column does indeed have no answers, so caution must be 
utilized. Another way in preventing the “zero columns” is recoding the original question at the very 
beginning (in this case, in SPSS before even running the results in SUDAAN). An example would 
be to recode the answer 8 to a 5.  
 

Charts 
After all this work was done and I proudly showed it to my supervisor, she said, “That’s nice! But 
what about some charts?” 
 
  So back to SAS Enhance Editor! (and the Help button!) 
 

The charts that are made in SAS are for internal use only; they do not go out to clients. We want 
to have a look at where the story is before we make charts for our clients in Excel.  

Section 14: 
 
proc sort data=present.charttest1; 

        by rf subtitle; 

run; 

 

data present.charttest2; 

        set present.charttest1; 

        by rf subtitle; 

    if first.subtitle then chartvar=0; 

    chartvar+1; 

run; 

 

The data are sorted by risk factor and the demographic titles. In the data step the X-plot line 
needed to be numbered in such a way that all the demographics for an area had the same 
number. So that, per the risk factor variable, every area and their demographics would have the 
same numbers throughout the data set. So if the risk factor had two answers, area one’s chartvar 
would have values 1 and 2 while area two’s chartvar would have values 3 and 4 and so forth. If 
the risk factor had four answers, area one’s chartvar would have values 1 through 4 and area two 
would have values 5 through 8. 
 
Picture 15a chartvar numbered one to sixteen for a given demographic: 

Picture 15b data sorted back to original setting: 

 
 
A chart template was found that is a bar chart that has upper and lower CI error markings. It took 
some time to find this and while not perfect, it works for our internal use. In SAS 9.3, much of this 
work will be probably be unnecessary.  
 
Section 15: 
proc template; 

        define statgraph brfsschart; 

         begingraph; 

entrytitle "Prevalence of Risk Factor by Demographics,  

   2010"; 

entryfootnote HALIGN=left "Error Bars show 95% Convidence 

 Intervals"; 

layout gridded / border=false; 

         layout datalattice columnvar=subtitle /     

      headerlabeldisplay=value cellwidthmin=20 

                columnheaders=bottom border=false columndatarange=union 

                columnaxisopts=(display=(line tickvalues)) 

                rowaxisopts=(offsetmin=0 linearopts=(viewmax=100   

    tickvaluepriority=true) 

                        label='Percentages' griddisplay=on); 

            layout prototype / walldisplay=(fill); 

              barchart x=chartvar y=rowper / group=area name='a'   

   outlineattrs=(color=black); 

            scatterplot x=chartvar y=rowper / yerrorlower=lowrow   

   yerrorupper=uprow MARKERATTRS=(size=0) 

   errorbarattrs=(thickness=1) datatransparency=0.6; 

            endlayout; 

        endlayout; 

        entry ' '; 

        discretelegend 'a' / title='Area: ' border=true; 

        endlayout; 

  endgraph; 

end; 

run; 

 
Most of this should be easy to figure out. Substitute the columns needed and then call the 
template. We first must resort the data set to their proper order after making the chartvar. 
 
Section 16: 
 
proc sort data=present.charttest2; 

        by area rf grp2; 

run; 

 

ods listing close; 

ods html image_dpi=100 file='presentcharts.xls' 

path='H:\Brfss\avincent\MyPresentations\'; 

ods graphics / reset noborder width=1000px height=500px 

               imagename='BRFSSChart' imagefmt=gif noscale; 

 

 

proc sgrender data=present.charttest2 template=brfsschart; 

  format area areaf. rf riskf.; 

  where (area=2 and rf=4 and chartvar=3); 

        by area rf; 

run; 

 

ods html close; 

ods listing; 

 
We use a ‘where’ statement in the proc sgrender (which is a proc step to identify the data set that 
contains the plot variables and the statgraph template) so that we don’t get all the answers all at 
once. 
 
And here are the results: 

Picture 16: 

 
 
The title didn’t put what the risk factor variable was, but that’s OK, it is internal use. The above 
table is for those who responded that they had diabetes.  
 
A rule of thumb about CI’s (technically statistical tests should be done), if the top of one bar 
overlaps the bottom of another bar then the two estimates are not statistically different. Compare 
the “male” and “female” error bars. They both overlap. Diabetes does not seem to be different 
between men and women. 
 
If they don’t over lap then they may be statistically significant by rule of thumb, but a good 
statistician would run a t-test or chi-test to be sure. In this case compare “black” with “white” or 
ages “18-29” to “65+” years. Both of these examples have bars that do not overlap  
 
Also the shorter the distance between upper and lower error bars the better, since there is less 
variability in the sample. Typically, larger samples have less variance. The longer bars, like the “< 
High School”, show less confidence in your estimate. 
 
Remember that SUDAAN does asymmetric confidence intervals, but SAS has symmetrical 
confidence intervals. So whenever anyone states that their percentages or estimates are “plus or 
minus” a certain number, do the math, see if they overlap!  

Conclusion 

This has been a little primer to show that anyone with just the basic knowledge of SAS can create 
some really nice reports just by looking at SAS help and SAS support online. So don’t be afraid to 
dive right in and play around till you get what you want. “Cut, paste, and tweak!” is my motto! 

Contact Information 

Comments or Questions: 
 
Anna Vincent 
Center for Health Statistics, MC-1898 
Texas Department of State Health Services 
PO Box 149317 
Austin, TX 78714-9347 
512-776-2520 work 
512-458-7429 fax 
Emails: 
Anna.vincent@dshs.state.tx.us 
brfss@dshs.state.tx.us 
Website: 
http://www.dshs.state.tx.us/chs/brfss/default.shtm  

